

```

1 ; Binärzähler
2
3     device 16f84
4
5 ; \***** labels *****\
6
7 ; Statusregister
8 status equ 3 ; Adresse des Statusregisters im RAM-File
9 rp0 equ 5 ; Bank select
10 carry equ 0 ; carry flag - Zeigt an, ob bei der letzten Rechenoperation ein Übertrag auftrat.
11 zero equ 2 ; zero flag - Zeigt an, ob das Ergebnis einer Operation gleich 0 war.
12
13 ; Port A bzw. 0
14 porta equ 5
15 ; Bitstellen von Port A
16 takt equ 0 ; Zähleingang RA0
17 reset equ 1 ; Reseteingang RA1
18 inhibit equ 2 ; Inhibit-Eingang RA2
19 carryOut equ 3 ; Carryausgang RA3
20 maske equ 1 ; 00000001 ; Maske für Zähleingang RA0
21
22 portb equ 6
23 ; TRIS: TRI-State. Mit einem TRIS-Register können Portpins auf Eingang oder Ausgang geschaltet werden.
24 trisa equ 5
25 trisb equ 6
26
27 ; Variablen - Speicherstellen
28 counter equ 10h ; 0c erste freie Adresse, 10h ist im Simulator leicht zu finden
29 aktWert equ 11h
30 alterWert equ 12h
31 flanke equ 13h
32
33 ; \***** labels *****\
34
35 org 0 ; Programm beginnt bei Adresse 0
36
37 cold
38     ; Ports initialisieren
39     bsf status,rp0 ; auf Bank 1 umschalten
40     bcf trisa,carryOut ; Port A carry auf Ausgang setzen
41     clrf trisb ; Port B alles Ausgang
42
43     bcf status,rp0 ; zurück auf Bank 0
44
45     ; ersten Wert lesen
46     movf porta,w ; Port A ins W-file lesen
47     andlw maske ; RA0 Zähleingang maskieren
48     movwf alterWert ; W-file in alterWert Schreiben: erster Vergleichswert
49
50 resetCNT
51     clrf counter ; Counter mit 0 initialisieren
52     bcf porta,carryOut ; carry zurücksetzen
53
54     ; Counterinhalt ausgeben
55     movf counter,w
56     movwf portb
57
58 mainloop
59     btfsc porta,reset ; Resteingang 1?
60     goto resetCNT ; ja → reset
61
62     btfsc porta,inhibit ; Inhibit-Eingang 1?
63     goto mainloop ; ja → Zähler anhalten
64
65     call checkEdge ; Flanke da? Nein → w = 0,
66     xorlw 2 ; setze zero-flag bei steigender Flanke w = 2

```

```

67     btfss status,zero ; w = 2?
68     goto mainloop ; nein
69
70     bcf porta,carryOut ; Ausgabe carry = 0
71     incf counter ; counter++
72     btfsc status,zero ; overflow?
73     bsf porta,carryOut ; ja
74
75     ; Ausgabe counter
76     movf counter,w
77     movwf portb
78
79     goto mainloop
80
81 checkEdge
82     ; aktuellen Wert lesen
83     movf porta,w ; Port A ins w-file lesen
84     andlw maske ; RA0 Zähleingang maskieren
85     movwf aktWert ; w-file in aktWert-File schreiben
86     xorwf alterWert,w ; mit alterWert vergleichen
87     movwf flanke ;
88
89     ; alterWert = aktWert
90     movf aktWert,w
91     movwf alterWert
92
93     ; flanke = 0 wenn aktWert = alterWert
94     movf flanke ; setze zero-flag wenn flanke = 0
95     btfsc status,zero ; flanke 0?
96     retlw 0 ; nein: keine neue Flanke
97
98     movf aktWert ; setze zero-flag wenn flanke = 0
99     btfss status,zero ; aktWert = 0? / zero = 1?
100    retlw 2 ; nein: steigende Flanke
101    retlw 1 ; ja: fallende Flanke

```